COMPUTATIONAL ECONOMICS

A General Equilibrium Model in MATLAB

Revised Edition

David A. Kendrick
Ruben P. Mercado
Hans M. Amman

1 introduction

The analysis of economy-wide models is a particularly demanding topic, as it involves the study of interdependence, which implies a move to the realm of multiple heterogeneous agents, sectors, and institutions interacting in complex ways. While there are some analytical methods and results available to help us in such endeavors, we have to resort to computational methods when we move on to medium- or large-size models or when we deal with particularly complex ones.

This chapter provides an introduction to the art of economy-wide modeling. We present a sequence of small models, show how to implement them in GAMS, perform some experiments, and suggest others. We start with an input-output model in which quantities produced are determined given technology and demand levels. We follow with a production prices model that determines relative prices given technology and a distributive variable. We then move on to a general equilibrium model in which prices and quantities are determined simultaneously given technology, preferences, and endowments. Finally, we introduce SAM-based and Johansen-style computable general equilibrium models. We present these models in a sequence that reflects primarily their computational complexity in terms of degree of nonlinearity and size. The sequence does not imply historical or theoretical precedence of one type of model over the others or a ranking of practical relevance.

2 Input-Output Model

A good starting point for the study of interdependence in economics is the well-known input-output model pioneered by Nobel prize winner Wassily Leontief (1953). This type of model was designed primarily for the determination of direct and indirect levels of production to satisfy a given increase in final demand.

Consider an economy with three industries (1, 2, and 3). Each of them produces a single output, using part of its own production as well as part of the output from the other industries as inputs. It is clear, then, that each industry plays a dual role since it is both a supplier of inputs and a user of outputs. Imagine that each product in this economy is also used to satisfy an exogenously given level of demand from consumers. In formal terms, we can represent the economy just described as follows[1]:

[1] One of the attractive features of input-output models is that in principle the data that are used to compute the coefficients in the model can be obtained directly from sources such as the

$$
\begin{aligned}
x_1 &= a_{11}\ x_1 + \ a_{12}\ x_2 + a_{13}\ x_3 + d_1 \\
x_2 &= a_{21}\ x_1 + \ a_{22}\ x_2 + a_{23}\ x_3 + d_2 \\
x_3 &= a_{31}\ x_1 + \ a_{32}\ x_2 + a_{33}\ x_3 + d_3
\end{aligned}
\tag{1}
$$

where

x_i = production levels

a_{ij} = the input-output coefficients (the intermediate requirements from industry i per unit of output of industry j

d_i = the levels of final demand from the consumers

In matrix notation, we can write equation (1) as

$$x = Ax + d \tag{2}$$

where

x = the vector of levels of production

d = the vector of final demands

A = the input-output coefficients matrix

A question can be posed for this economy. Given an example input-output coefficients matrix,

$$A = \begin{bmatrix} 0.3 & 0.2 & 0.2 \\ 0.1 & 0.4 & 0.5 \\ 0.4 & 0.1 & 0.2 \end{bmatrix} \tag{3}$$

and an example vector of final demands,

$$d = \begin{bmatrix} 4 \\ 5 \\ 3 \end{bmatrix} \tag{4}$$

what is the required level of total production of each industry (direct and indirect) to satisfy that final demand vector? The GAMS representation of this problem is:

manufacturing censuses done in many countries.

```
$TITLE IO-1
* Input-Output Model
SCALARS

d1 final demand for x1 /4/
d2 final demand for x2 /5/
d3 final demand for x3 /3/;

VARIABLES

x1 production level industry 1
x2 production level industry 2
x3 production level industry 3

j performance index;

EQUATIONS

eqx1
eqx2
eqx3

jd performance index definition;

jd.. j =E= 0;

eqx1.. x1 =E= 0.3*x1 + 0.2*x2 + 0.2*x3 + d1;
eqx2.. x2 =E= 0.1*x1 + 0.4*x2 + 0.5*x3 + d2;
eqx3.. x3 =E= 0.4*x1 + 0.1*x2 + 0.2*x3 + d3;

MODEL IO /jd, eqx1, eqx2, eqx3/;
SOLVE IO MAXIMIZING J USING LP;
DISPLAY x1.l, x2.l, x3.l;
```

The GAMS files for this and the other models in this chapter are in the book web site under the names listed in each $TITLE statement. Note that in this model, as discussed in Appendix A, in order to solve a system of simultaneous equations in GAMS we must include an additional variable and equation. As shown in that appendix, GAMS has procedures for optimizing but not for solving simultaneous equations. Therefore, the method for solving simultaneous equations in this software system is to add an additional variable to the model—j in this case—and an additional equation—jd in this case. Then the additional variable is maximized or minimized in order to find the solution to the model. The solution obtained using this method is

$$x_1 = 16.821, \quad x_2 = 23.744, \quad x_3 = 15.128.$$

There are analytical methods available to deal with this problem.[2] Indeed, the analytical solution is obtained by solving equation (2) to obtain

$$x = (I - A)^{-1} d \tag{5}$$

where I is the identity matrix.[3] This formula can be easily handled for small models. However, as soon as one moves to larger models computational methods are required to perform the matrix inversion, and these methods become unavoidable as we move to more complex problems. For example, imagine now that we have some restriction, such as a capacity constraint, on the maximum level of production of some products (say $x_2 \leqslant 22$ and $x_3 \leqslant 14$) and we want to know the maximum level of final demand of Product 1 (d_1) that the economy can satisfy, given the final demand levels d_2 and d_3. This can be easily handled in GAMS. Here is the corresponding GAMS representation of the problem:

```
\begin{verbatim}
$TITLE IO-2
* Input-Output Model with restrictions

SCALARS

d2 final demand for x2 /5/
d3 final demand for x3 /3/;

POSITIVE VARIABLES

x1 production level industry 1
x2 production level industry 2
x3 production level industry 3
d1 final demand for x1;

VARIABLES

j performance index;

EQUATIONS

eqx1
eqx2
eqx3

res1 restriction 1
res2 restriction 2
```


[2] See for example Chiang (1974) for an introduction to these methods.
[3] Also, it is necessary that the I-A matrix be nonsingular.

```
jd performance index definition;

jd.. j =E= d1;

eqx1.. x1 =E= 0.3*x1 + 0.2*x2 + 0.2*x3 + d1;
eqx2.. x2 =E= 0.1*x1 + 0.4*x2 + 0.5*x3 + d2;
eqx3.. x3 =E= 0.4*x1 + 0.1*x2 + 0.2*x3 + d3;
res1.. x2 =L= 22;
res2.. x3 =L= 14;

MODEL IO /all/;
SOLVE IO MAXIMIZING j USING LP;
DISPLAY x1.l, x2.l, x3.l, d1.l;
```

Note that we define and add two equations (res1 and res2) corresponding to the restrictions, set the performance index j equal tod_1, and define d_1 as a variable (no longer as a scalar). In addition, to avoid negative values that make no economic sense we define all variables except the performance index as positive. Solving the problem, we obtain

$$x_1 = 14.143, \quad x_2 = 22, \quad x_3 = 13.571, \quad d_1 = 2.786$$

in contrast with our original solution of

$$x_1 = 16.821, \quad x_2 = 23.744, \quad x_3 = 15.128, \quad d_1 = 4$$

Thus the level of final demand for Good 1 is lower once the restrictions are in place and we can achieve only 2.786. This is lower than in the original case since we set the values of the restrictions below the solution levels previously obtained. On the other hand, if the economy is able to lift those *bottlenecks* up to 30 for x_2 and 20 forx_3, the demand for goods produced by Sector 1 that could be satisfied would be $d_1 = 7.8$.

3 Production Prices Model

So far we have been dealing with a model with two main types of agents (consumers and industries), in which their interrelations are linear and where, given a technology (the input-output coefficients matrix), we determine quantities produced and/or demanded. Implicitly, relative prices are taken as given. We move now to a nonlinear model in which prices are determined given technology and a distributive variable. This type of model was pioneered by Ricardo (1817) at the beginning of the nineteenth century and later formalized by Sraffa

(1972). One of its main goals is to allow us to study issues of income distribution between wages and profits.

Let us define

$$
\begin{aligned}
v &= \text{value of intermediate inputs} \\
\pi &= \text{profit} \\
w &= \text{wage cost} \\
p &= \text{price}
\end{aligned}
$$

We can then write

$$v + \pi + w = p \tag{6}$$

This equation simply requires that the total cost—that is, the sum of the three elements of cost, namely intermediate goods, capital, and labor—be equal to the price. Then assuming that profits are equal to the profit rate r times the value of the intermediate inputs, we have

$$v + v\,r + w = p \tag{7}$$

or

$$v\,(1 + r) + w = p \tag{8}$$

Then using the input-output coefficients for the intermediate inputs, a simple three-good production prices model can be formalized as:

$$
\begin{aligned}
(a_{11}\ p_1 + \ a_{21}\ p_2 + a_{31}\ p_3)\ (1+r) + l_1 w &= p_1 \\
(a_{12}\ p_1 + \ a_{22}\ p_2 + a_{32}\ p_3)\ (1+r) + l_2 w &= p_2 \\
(a_{13}\ p_1 + \ a_{23}\ p_2 + a_{33}\ p_3)\ (1+r) + l_3 w &= p_3
\end{aligned}
\tag{9}
$$

The a's are, as before, input-output coefficients. Note that subscripts on these coefficients are reversed, that is, the input-output matrix is the transpose of the A matrix corresponding to input-output Leontief-type models. This is so because here we determine prices given technology, whereas in Leontief models we determine quantities given technology.[4] The l's are also input-output coefficients indicating the quantity of labor required for the production of one unit of product. In addition, the p's are relative prices, w is the wage per unit of labor (assumed to be uniform for the whole economy), and r is the profit rate. The profit rate is the same for every industry, implying that we are dealing with a

[4] To learn more about this, see Passinetti (1977).

long-run situation in which capital earns the same profit no matter the industry. Otherwise there would be capital movements from industries with a low rate to those with a higher rate until that rate equalizes across industries.

The foregoing model has five variables and three equations. Since all prices are relative prices, we have to choose one of them as a numeraire in order for all the other pricelike variables to be expressed in its terms. We can do this by fixing one variable (say, one price).[5] Once we have done this, we are still left with a degree of freedom regarding w and r to close the system of equations, and we can fix, for example, the wage w.[6]

In the GAMS representation of this model that follows, we have chosen a particular set of values for the input-output coefficients, and have set $p_1 = 1$ and $w = 0$.

```
$TITLE ProdPri
* Production Prices Model

SCALARS

L1 /0.2/
L2 /0.5/
L3 /0.3/;

VARIABLES

p1
p2
p3
w
r

j performance index;

EQUATIONS
```


[5] For Sraffa, the choice of the numeraire involved other issues dating back to Ricardo. Facing a change in the relative price of a commodity, Ricardo wanted to be able to tell when the change originated in the conditions affecting the production of that commodity or in the conditions of production of the commodity being used as numeraire. To solve in part that problem, Sraffa built a numeraire that takes the form of a restriction involving some of the model variables. This is a complex theoretical issue and we do not deal with it here. See Sraffa (1972).

[6] Classical economists like Ricardo used to consider that w was determined by the minimum subsistence level of the labor force. More modern approaches have considered that w is the outcome of the bargaining process between workers' unions and industrialists' unions.

```
eqp1
eqp2
eqp3

jd performance index definition;

jd.. j =E= 0;

eqp1.. (0.3*p1 + 0.1*p2 + 0.4*p3) * (1+r) + L1 * w =E= p1;
eqp2.. (0.2*p1 + 0.4*p2 + 0.1*p3) * (1+r) + L2 * w =E= p2;
eqp3.. (0.2*p1 + 0.5*p2 + 0.2*p3) * (1+r) + L3 * w =E= p3;

w.fx = 0;
p1.fx = 1;

MODEL PP1 /all/;
SOLVE PP1 MAXIMIZING J USING NLP;
DISPLAY p1.l, p2.l, p3.l, w.l, r.l;
```

Note that the statements

```
w.fx = 0;
p1.fx = 1;
```

are used to fix w and p1.The solution for r is 0.25. It is interesting to observe what happens as we decrease r. To do so, we now set $r$equal to different fixed values, that is, we substitute r.fx = 0.25 (and later r.fx=0.20, and so on) for w.fx = 0 in the foregoing GAMS representation. We find that there is an inverse relationship between the wage w and the profit rate r, such as the one shown in Table 8.1.

Table 1: Wages and Profits

r	w
0.25	0.000
0.20	0.157
0.15	0.270
0.10	0.389
0.05	0.515
0.00	0.648

In this example, not only wages, but also prices go up as $r$decreases., but, in general, prices can go either way—some may go up, others down. However, if we choose w as the numeraire, we observe that as r increases, all prices increase, indicating that the real wage decreases no matter what weights are used to compute the corresponding wage deflator.

4 General Equilibrium Model

In the previous two sections we considered first a quantity model and then a price model. Here we move on to a model in which quantities and prices are determined simultaneously. General equilibrium models of this type were pioneered by Leon Walras (1874) and generalized by Nobel Prize winners Kenneth Arrow (1971) and Gerard Debreu (1986). One of the main goals of general equilibrium modeling is the study of changes in prices and quantities when technology, preferences, or endowments change.

Imagine that we have a very simple economy, with only one production sector, two factors of production, and a single household. The production sector produces a single good q_s (output supply) with a Cobb-Douglas constant returns to scale production technology using two inputs: labor and capital. Technical progress (b) can affect total factor productivity. The corresponding labor and capital demand functions (l_d and k_d) are derived by combining the production function with the assumption of profit-maximizing behavior. Labor and capital supplies ($l_s$and k_s) are given exogenously. The single household provides labor and capital in exchange for the corresponding wage (w) and profit (r), spending all its income (y) in the demand for the single good (q_d). So far, we have three markets: labor, capital, and good markets, and we impose market-clearing conditions specifying that supply equals demand.

To obtain the model euations, we maximize the profit function

$$\pi = p\, q_s - w\, l_d - r\, k_d \tag{10}$$

subject to the production function

$$q_s = b\; l_d^a\, k_d^{1-a} \tag{11}$$

Substituting the production function into the profit function, the first-order conditions are

$$\frac{\partial \pi}{\partial l_d} = p\, a\, b\, l_d^{a-1}\, k_d^{1-a} - w = 0 \tag{12}$$

$$\frac{\partial \pi}{\partial k_d} = p\ (1-a)\, b\, l_d^a\, k_d^{-a} - r = 0 \tag{13}$$

Substituting the production function into (12) and (13) and rearranging terms we obtain, respectively, the labor and capital demand functions:

$$l_d = \frac{a\, q_s\, p}{w} \tag{14}$$

and

$$k_d = \frac{(1-a)\, q_s\, p}{r} \tag{15}$$

Production function (Cobb-Douglas):

$$q_s = b\ \ l_d^a\, k_d^{1-a}$$

Labor demand, supply, and market clearing:

$$l_d = \frac{a\ q_s p}{w}$$

$$l_s = \bar{l}_s$$

$$l_s = l_d$$

Capital demand, supply, and market clearing:

$$k_d = \frac{(1-a)\ q_s p}{r}, k_s = \bar{k}_s, k_s = k_d$$

Household income:

$$y = w\ l_d + r\ k_d$$

Good demand:

$$q_d = \frac{y}{p}$$

Good market clearing:

$$q_s = q_d$$

This simple model has ten variables and ten equations. However, one of them is redundant, since Walras's law establishes that for n-markets we need only $n - 1$ equilibrium conditions. Moreover, since this model determines relative prices (p, w, and r), we have to fix one of them as the numeraire. Thus, by choosing one price as the numeraire (say we fix $p = 1$) and deleting the corresponding good market-clearing equation ($q_s = q_d$), we are left with a nine-variable nine-equation well-defined model. We do not consider the performance index j (used in the GAMS representation that follows) in the variable count nor the performance index definition in the equation count.

The GAMS representation of the model follows. Arbitrary, but reasonable, numbers have been chosen for the parameters and for the labor and capital stocks:

```
$TITLE SIMPLEGE

SCALARS

a labor share / 0.7 /
b technology parameter / 1.2 /;

POSITIVE VARIABLES

qs good supply
qd good demand
ld labor demand
ls labor supply
kd capital demand
ks capital supply

p price
w wage
r profit
y income;

VARIABLES

j performance index;

EQUATIONS

eqs good supply equation (production function)
eqd good demand equation
eld labor demand equation
els labor supply equation
ekd capital demand equation
eks capital supply equation
```

```
ey income equation
eml labor market clearing
emk capital market clearing

jd performance index definition;

jd.. j =E= 0;

eqs.. qs =E= b * ld**a * kd**(1-a);
eld.. ld =E= a * qs * p / w;
els.. ls =E= 2;
eml.. ld =E= ls;
ekd.. kd =E= (1-a)* qs * p / r;
eks.. ks =E= 1;
emk.. kd =E= ks;
ey.. y =E= w * ld + r * kd;
eqd.. qd =E= y / p;

*lower bounds to avoid division by zero

p.lo = 0.001; w.lo = 0.001; r.lo = 0.001;

*numeraire

p.fx = 1;

MODEL SIMPLEGE /all/;
SOLVE SIMPLEGE MAXIMIZING J USING NLP;
DISPLAY qs.l, qd.l, ld.l, ls.l, kd.l, ks.l, p.l, w.l, r.l, y.l;

The solution values are:

qs.L = 1.949 good supply
qd.L = 1.949 good demand
ld.L = 2.000 labor demand
ls.L = 2.000 labor supply
kd.L = 1.000 capital demand
ks.L = 1.000 capital supply
p.L = 1.000 price
w.L = 0.682 wage
r.L = 0.585 profit
y.L = 1.949 income
```

It is important to perform some basic checks on the workings of the model. For instance, since we assumed market clearing, we have to verify that supply equals demand in each market. In addition, when increasing the value of the numeraire, all quantity variables should remain the same, while nominal variables (prices and income) should increase proportionally.

Note that this model, as the others introduced previously, are models of the *real* side of the economy, in the sense that they do not include money explicitly. Moreover, the result that real variables remain the same while nominal variables change in proportion to the numeraire can be interpreted as meaning that in this model money is neutral.

5 Computable General Equilibrium Models

So far we have presented very small models. However, applied economy-wide models tend to be large, so that the use of computational techniques becomes unavoidable. In this section we introduce a slightly larger model than the general equilibrium model presented in Section 8.3, to provide a flavor of what is like to deal with more than a handful of variables and equations. Models like this are known in the literature as computable general equilibrium (CGE) models. Later we go back to a small model to illustrate the application of a linearization technique that is useful when dealing with relatively large nonlinear models. Note that the material in the remainder of this chapter is considerably more difficult than in the previous sections, and that the exposition moves at a more rapid pace.

5.1 A SAM-Based Model

We move now to a two-sector, two-factor, and two-household model to illustrate how to build a CGE model based on a Social Accounting Matrix (SAM). This model was developed by Arne Drud at the World Bank and is discussed in Kendrick (1990).[7]

Based on the research of Nobel Prize winner Richard Stone (1961), a SAM contains information on the flow of goods and payments between institutions in the economy. In Table 2 we present a simple SAM, where the table should be

[7] Drud implemented the model in Hercules, a system that allowed one to develop CGE models by providing basic information in the form of SAMs and by choosing from a menu the functional forms for production and demand functions. Hercules is no longer in use; however, GAMS now provides a solver (MPSGE) that performs similar functions to those of Hercules (see `www.gams.com`). These types of systems for model representation are very useful and especially time saving for the experienced modeler. However, here we present a direct GAMS representation of the Drud model, which is more suitable for introducing beginners to basic issues in computational model building.

read following the principle that columns pay rows and where each column adds up to the same number as the corresponding row. For example, the food industry pays 75 to labor and 50 to capital. Labor pays 90 to rural households and 70 to urban households. Urban households spend 65 on food and 85 on clothing.

Table 2: A Simple SAM

	Factors	**Households**	**Sectors**
	Labor Capital	Rural Urban	Food Clothing
Factors Labor Capital			1. 85 50 60
Households Rural Urban	1. 30 70 80		
Sectors Food Clothing		1. 65 60 85	

Usually, a SAM can be constructed using a country's official statistics such us the national accounts. Based on Table 8.2, Drud built the model shown in Table 8.3.

The model contains three key types of variables: price (p), quantity (q), and income (y), all of them with a single subscript since they apply to a single institution (subscript f indicates factor, h household, and s sector). There are also two additional types of variables: payment (t) and commodity (c), with two subscripts since they represent flows of goods and payment. The subscripts on the payment variables t follow the SAM convention: payments are from columns to rows (i.e., t_{fs} indicates payment from sector s to factor f). Commodity flows (c) follow the more common forward subscript convention (i.e., c_{fs} indicates the flow of factor f to sector s, while c_{sh} is the flow of purchased goods from sector s to household h).

Table 3: Drud's Model

	Quantity q	Price, share, or payment p	Price-quantity pq
Sectors Output Input	$q_s = b_s \prod_f c_{fs}^{a_{fs}}$ $c_{fs} = \frac{a_{fs}\, q_s\, p_s}{p_f}$		$y_s = p_s\, q_s$ $t_{fs} = p_f\, c_{fs}$
Factors Income Transfer		$t_{hf} = a_{hf}\, q_f$	$y_f = p_f\, q_f$
Household Consumption CPI		$t_{sh} = a_{sh}\, q_h$ $p_h = \prod_s p_s^{a_{sh}}$	$t_{sh} = p_s\, c_{sh}$ $y_h = p_h\, q_h$
Linkage Sectors Factors Households		$y_s = \sum_h t_{sh}$ $y_f = \sum_s t_{fs}$ $y_h = \sum_f t_{hf}$	

The output-quantity equations specify production functions with a Cobb-Douglas technology, where b is a technology parameter. The input-quantity equations are the corresponding factor-demand equations derived from the production functions and imposing a zero-profit condition. The CPI-price equations are price indices for the rural and urban households, respectively. The a's are share parameters derived from the SAM.

When expanded, the model has thirty-eight variables and thirty-six equations. Take the amount of labor and capital as given (i.e., as exogenous variables). Choose one price as the numeraire (say we fix $p_{(urban)} = 1$). Delete the corresponding market-clearing equation (in this case, deleting the linkage equation $y_{(urban)} = \sum_{f} t_{(urban,f)}$ does the job). Then we are left with a model with thirty-six endogenous variables and thirty-six equations. The GAMS representation of this model is as follows:

```
$TITLE SAM-Drud-Kendrick
* Developed by Ruben Mercado

options limrow = 4;
* option NLP=MINOS;
* Note to PC users-- From the proceeding line, remove the asterisk.
* i.e., on a PC, this code must be run with the option NLP=MINOS.

SETS
i general index /labor, capital, rural, urban, food, clothing/
s(i) sectors /food, clothing/
f(i) factors /labor, capital/
h(i) households /rural, urban/;
ALIAS (i,ip);
ALIAS (i,iq);

PARAMETERS
b(s) technical coefficients
a(i,ip) share coefficients;
b('food') = 1.2;  b('clothing') = 1;

TABLE sam(i,ip)
           labor capital rural urban food clothing
labor                                75   85
capital                              50   60
rural        90     30
urban        70     80
food                     60     65
clothing                 60     85             ;

a(i,ip)= sam(i,ip) / sum(iq, sam(iq,ip));
```

```
DISPLAY a;

POSITIVE VARIABLES
p(i) price
q(i) quantiy
y(i) income
t(i,ip) payment
c(i,ip) commodity ;

VARIABLES
j  performance index;

EQUATIONS
eph(h)
eqs(s)
eys(s)
eyf(f)
eyh(h)
etfs(f,s)
ethf(h,f)
etsh(s,h)
eetsh(s,h)
ecfs(f,s)
eeys(s)
eeyf(f)
eeyh(h)
jd performance index definition;
* performance index equation
jd..    j =E= 0;

*sectors
eqs(s)..      q(s) =E= b(s)* prod(f, c(f,s)**a(f,s));
ecfs(f,s).. c(f,s) =E= a(f,s) * q(s) * p(s) / p(f);
eys(s)..      y(s) =E= p(s) * q(s);
etfs(f,s).. t(f,s) =E= p(f) * c(f,s);

*factors
eyf(f)..      y(f) =E= p(f) * q(f);
ethf(h,f)..  t(h,f)=E= a(h,f) * y(f);

*households
etsh(s,h)..       t(s,h) =E= a(s,h) * y(h);
eph(h)..             p(h)=E= prod(s, p(s)**a(s,h));
eetsh(s,h)..       t(s,h)=E= p(s) * c(s,h);
eyh(h)..            y(h) =E= p(h) * q(h);

*linkage
eeys(s)..             y(s) =E= sum(h,t(s,h));
eeyf(f)..             y(f) =E= sum(s,t(f,s));
```

```
eeyh('rural').. y('rural') =E= sum(f,t('rural',f));
*notice that we eliminate one of the linkage equations (Walras law)

*initial values to facilitate solver convergence
p.l(i) = 1;  q.l(i) = 1;  y.l(i) = 1;

*lower bound to avoid division by zero
p.lo(f) = 0.001;

*lower bounds to avoid undefined derivatives in exponential functions
p.lo(s) = 0.001;  c.lo(f,s) = 0.001;

*exogenous variables
q.fx('labor') = 2;  q.fx('capital') = 1;

*numeraire
p.fx('urban') = 1;

MODEL SAMDK /all/;
option iterlim = 10000;
SOLVE SAMDK MAXIMIZING J USING NLP;

PARAMETER REPORT;
REPORT(i, "price") = p.l(i);
REPORT(i, "quantity") = q.l(i);
REPORT(i, "income") = y.l(i);

DISPLAY REPORT;  DISPLAY t.l, c.l;
```

The GAMS representation is similar to the simple general equilibrium model presented earlier. Here we make use of sets and subsets as indices; we use the ALIAS command to redefine an index so we can use it to index a matrix; we input the SAM as a table under the PARAMETER section; and we define indexed variables and equations. Note that in order to have a more compact representation, we were able to use a general index i for variables, and later work with subsets of variables, but we did not do so for equations. GAMS does not admit the use of subsets as indices of equations.

As in the previous example, we should check that only nominal variables change (proportionally) when we change the numeraire.

5.2 A Johansen-Style Model

CGE models tend to be large and nonlinear. As they grow in size, obtaining convergence (i.e., a numerical solution) is likely to become more difficult. An alternative is to switch to a model representation pioneered by Leif Johansen (1960). Johansen-style models are solved in a linearized form where all the variables are rates of growth. This method consists in transforming all the variables in the model into percentage changes with respect to a base case.

For example, given an expression in levels like

$$X = a\ Y\ Z \tag{16}$$

if we first take logs, we obtain

$$\log X = \log a + \log\ Y + \log\ Z \tag{17}$$

and totally differentiating,

$$d(\log X) = d(\log a) + d(\log\ Y) + d(\log\ Z) \tag{18}$$

that is (since a is a constant),

$$\frac{dX}{X} = \frac{dY}{Y} + \frac{dZ}{Z} \tag{19}$$

or

$$x = \ y + z \tag{20}$$

where x, y, and z variables are percentage deviations. An alternative derivation without using logs is to use the total differential

$$dX = YZda + aZdY + aYdZ \tag{21}$$

$$dX = aZdY + aYdZ \tag{22}$$

$$\frac{dX}{X} = \frac{aZdY}{X} + \frac{aYdZ}{X} \tag{23}$$

$$\frac{dX}{X} = \frac{dY}{Y} + \frac{dZ}{Z} \tag{24}$$

which leads to the ame result. In a similar fashion, we can transform

$$X = a\ Y^b \tag{25}$$

into

$$x = b\ y \tag{26}$$

Thus for an expression like

$$X = Y + Z \tag{27}$$

we totally differentiate

$$dX = dY + dZ \tag{28}$$

then divide by the right-hand-side variable

$$\frac{dX}{X} = \frac{dY}{X} + \frac{dZ}{X} \tag{29}$$

We then multiply and divide each term on the right-hand side by the variable in its numerator and rearrange to obtain

$$\frac{dX}{X} = \frac{dY}{Y}\frac{Y}{X} + \frac{dZ}{Z}\frac{Z}{X} \tag{30}$$

or

$$\frac{dX}{X} = \frac{Y}{X}\frac{dY}{Y} + \frac{Z}{X}\frac{dZ}{Z} \tag{31}$$

or

$$x = s_y y + s_z z$$

where s_y and s_z are the shares

$$s_y = \frac{Y}{X} = \frac{Y}{Y+Z} \tag{32}$$

and

$$s_z = \frac{Z}{X} = \frac{Z}{Y+Z} \tag{33}$$

In short, the transformation of a model in levels into one in percentage changes can, in many cases, be achieved by applying some simple rules. Given X, Y, and Z as variables in levels, $a$and b as parameters and x, y, and z as variables in percentage deviations, some useful rules are

$$X = a\, Y\, Z \tag{34}$$

becomes

$$x = \ y + z \tag{35}$$

$$X = a\,Y^b \tag{36}$$

becomes

$$x = b\,y \tag{37}$$

$$X = Y + Z \tag{38}$$

becomes

$$x = s_y y + s_z z \tag{39}$$

where s_y and s_z are the shares $s_y = Y/(Y+Z)$ and $s_z = Z/(Y+Z)$.

Applying these rules to the simple general equilibrium model presented in Section 8.3 and interpreting each variable not as levels but as percentage changes with respect to a base case, we obtain the following GAMS representation:

```
$TITLE JohansenGE
* Developed by Ruben Mercado

SCALARS
a labor share / 0.7 /

VARIABLES
qs good supply
qd good demand
ld labor demand
ls labor supply
kd capital demand
ks capital supply
p price
w wage
r profit
y income
j  performance index;

EQUATIONS
eqs good supply equation (production funcion)
eqd good demand equation
eld labor demand equation
els labor supply equation
ekd capital demand equation
eks capital supply equation
ey  income equation
eml labor market clearing
```

```
emk capital market clearing
jd performance index definition;

jd..       j =E= 0;

eqs..     qs =E= ld * a + kd *(1-a);
eld..     ld =E= qs + p - w;
els..     ls =E= 0;
eml..     ld =E= ls;
ekd..     kd =E= qs + p - r;
eks..     ks =E= 0;
emk..     kd =E= ks;
ey..       y =E= (0.7)*(w + ld)+ 0.3 *(r + kd);
eqd..     qd =E= y - p;

*numeraire
p.fx = 0;

MODEL JOHANSENGE /all/;
SOLVE JOHANSENGE MAXIMIZING J USING LP;
DISPLAY qs.l, qd.l, ld.l, ls.l, kd.l, ks.l, p.l, w.l, r.l, y.l;
```

Note that we eliminated the b parameter from the scalars section, since we do not use it here. Furthermore, since percentage changes can be positive or negative, we no longer define the model variables as positive variables as we did in the version of the model where variables were in levels. Finally, the values of the stock of labor and capital and the numeraire are equal to zero, since they are percentage changes. The 0.7 and 0.3 coefficients that appear in equation `ey` are the corresponding share parameters obtained when applying the third rule. Finally, we solve the model invoking a linear programming solver, since the problem is a linear one.

An interesting exercise is to compare the results of the nonlinear model in levels versus the linear model in percentage changes for a given change in an exogenous variable. For example, say we increase the stock of capital by 20 percent. This means that in the nonlinear model k goes from 1 to 1.2, whereas in the linear model it goes from 0 to 0.2. The results are shown in Table 8.4.

The differences between the last two columns give us an idea of the approximation error of the linearized solution. We should expect this error to be larger the greater the change in the exogenous variables. Furthermore, note that if we simultaneously change the value of more than one exogenous variable for the linear version, the superposition principle applies: the combined effect of changes in more than one exogenous variable is equal to the sum of the individual effects.

Table 4: Comparison of Nonlinear and Linearized Models

	Nonlinear model			Linearized model
Variable	Solution $k = 1$	Solution $k = 1.2$	Percentage change	Percentage change
q	1.949	2.059	5.6	6
l	2	2	0	0
k	1	1.2	20	20
w	0.682	0.721	5.7	6
r	0.585	0.515	–12	–14
y	1.949	2.059	5.6	6

As we said earlier, solving nonlinear models may become problematic as they grow in size. The problem we just linearized using Johansen's technique is a very small one, and we used it to provide a simple illustration of the methodology. For an application to a larger model see Kendrick (1990), which provides a Johansen-style GAMS representation of a version of the ORANI model developed by Dixon et al. (1982) in Project Impact in Australia.

6 Experiments

For the input-output model in Section 8.1 you may perform experiments changing the levels of final demand, the values of some input-output coefficients or the nature of the capacity constraint restrictions.

For the production prices model in Section 8.2, an interesting experiment would be to pick one price as the numeraire (say $p_1 = 1$) and a technology such that the proportion between labor costs and total input costs is the same for each industry, that is, when the input-output coefficients are proportional for all industries. For instance, when the input-output matrix is

$$A = \begin{bmatrix} 0.05 & 0.025 & 0.1 \\ 0.1 & 0.05 & 0.2 \\ 0.2 & 0.1 & 0.4 \end{bmatrix} \tag{40}$$

and the labor coefficients vector is

$$L = \begin{bmatrix} 1/7 \\ 2/7 \\ 4/7 \end{bmatrix} \tag{41}$$

you observe that prices do not change as r and w change in an inverse relationship.

For the small general equilibrium model in Section 8.3 the economy-wide effects of technological progress can be simulated by increasing the value of the b parameter. You might also change the supply of labor or of capital and see how the wage and the profit levels are affected. If you do so, you observe that quantities do not change, only the wage and the profit rate change. Quantities would change if you specified elastic labor and capital supply functions, instead of the fixed supplies assumed in the model. Also, we imposed the market-clearing condition in all three markets. However, it may well be the case that such a condition is not appropriate for some markets because they are in disequilibrium. That may happen, for example, because their prices are exogenously fixed. For such cases we should follow an appropriate modeling strategy, such as the ones proposed, for example, by Malinvaud (1977).

Finally, for the SAM-based CGE model in Section 4.1, you can perform interesting experiments by changing the amount of labor or capital or the technology parameters. Note that you could also change the share parameters by changing some numbers in the SAM. If you do so, remember to maintain the corresponding balance between rows and columns. Another interesting exercise would be to expand the model to incorporate foreign trade as in Kendrick (1990).

7 Further Reading

Dervis et al. (1982) and Dixon, Powell, Parmenter, and Wilcoxen (1992) provide extended textbook presentations of the different types of models introduced in this chapter. For historical and analytical presentations of input-output and production prices models see Pasinetti (1977) and for CGE models see Dixon and Parmenter (1996). Shoven and Whalley (1992) deal extensively with neoclassical-type CGE models, while Taylor (1990) presents neostructuralist-type CGE models. Roland-Holst, Reinert, and Shiells (1994) provide an analysis of the North American Free Trade Area. Lofgren, Lee Harris, and Robinson (2002) develop a standard CGE model in GAMS. For the use of a dynamic CGE model in a control context to study income distribution changes, see Paez

(1999). For an approach to solving dynamic CGE models with stochastic control theory methods see Kim (2004).

References

Arrow, K. and Hahn, F.: 1971, *General Competitive Analysis*, Holden-Day, San Francisco, California, USA.

Chiang, A. C.: 1974, *Fundamental methods of mathematical economics*, International Student Edition, 2nd edn, McGraw-Hill Kogakusha, Tokyo, Japan.

Debreu, G.: 1986, *Theory of Value: An Axiomatic Analysis of Economic Equilibrium*, Yale University Press, New Haven, Connecticut, USA.

Dervis, K., de Melo, J. and Robinson, S.: 1982, *General Equilibrium Models for Development Policy*, Cambridge University Press, Cambridge, United Kingdom.

Dixon, P. B. and Parmenter, B. R.: 1996, Computable general equilibrium modeling for policy analysis and forecasting, *in* H. M. Amman, D. A. Kendrick and J. Rust (eds), *Handbook of Computational Economics*, Vol. 1 of *Handbook in Economics*, North Holland, Amsterdam, the Netherlands, pp. 3–86.

Dixon, P. B., Parmenter, B. R., Powell, A. A. and Wilcoxen, P. J.: 1992, *Notes and Problems in Applied General Equilibrium Economics*, North Holland, Amsterdam, the Netherlands.

Dixon, P. B., Parmenter, B. R., Sutton, J. and Vincent, D. P.: 1982, *ORANI, A Multisectoral Model of the Australian Economy*, North Holland, Amsterdam, the Netherlands.

Johansen, L.: 1960, *A Multi-Sectoral Model of Economic Growth*, North-Holland, Amsterdam, The Netehrlands.

Kendrick, D. A.: 1990, *Models for Analyzing Comparative Advantage*, Kluwer Academic Publishers, Dordrecht, the Netherlands.

Kim, S. R.: 2004, Uncertainty, political preferences, and stabilization: Stochastic control using dynamic cge models, *Computational Economics* **24**, 97–116.

Leontief, W.: 1953, *Studies in the Structure of the American Economy*, Oxford University Press, Oxford, United Kindom.

Lofgren, H. R., Harris, L. and Robinson, S.: 2002, A standard computable general equilibrium (cge) model in gams, *Microcomputers in Policy Research 5*, International Food Policy Institute, Washington, D.C., USA.

Malinvaud, E.: 1977, *The Theory of Unemployment Reconsidered*, Basil Blackwell, Oxford, United Kingdom.

Paez, P.: 1999, *An Optimal Control Framework for the Design of Alternative Macroeconomic Policies*, PhD thesis, University of Texas, Austin, Texas, USA.

Passinetti, L.: 1977, *Lectures on the Theory of Production*, Columbia University Press, New York, USA.

Ricardo, D.: 1817, *Principles of Political Economy and Taxation*, Dover Books.

Roland-Holst, D. W., Reinert, K. A. and Shiells, C. R.: 1994, A general equilibrium analysis of north american economic integration, *in* J. F. Francois and C. R. Shiells (eds), *Modeling Trade Policy: Applied General Equilibrium Assessments of North American Free Trade*, Cambridge University Press, Cambridge, Massachussets, USA, pp. 47–82.

Shoven, J. and Whalley, J.: 1992, *Applying General Equilibrium*, Cambridge University Press, Cambridge, United Kingdom.

Sraffa, P.: 1972, *Production of Commodities by Means of Commodities*, Cambridge University Press, Cambridge, United Kingdom.

Stone, J. R. N.: 1961, *Input-Output National Accounts*, OECD, Paris, France.

Taylor, L.: 1990, *Socially Relevant Policy Analysis: Structuralist Computable General Equilibrium Models for the Developing World*, The MIT Press, Cambridge, Massachussets, USA.

Walras, L.: 1874, *Elements of Pure Economics*, George Allen and Unwin (1954), London, United Kingdom.

Appendix

A The Stacking Method in GAMS

As a compact way of expressing a multiequation model, GAMS allows us to write indexed equations. As seen in several of the chapters in this book, those indices may represent commodities, locations, time periods, and so on.

For example, the equations corresponding to a problem such as

$$\max J = \sum_{i=0}^{2} w_1 x_i + w_2 y_i \tag{A-1}$$

subject to the constraints

$$a_{11} x_i + a_{12} y_i = b_1 \tag{A-2}$$
$$a_{21} x_i + a_{22} y_i = b_2 \tag{A-3}$$

can be represented in GAMS as

```
eqj..       j =e= sum(i, w1 * x(i) + w2 * y(i));
eq1(i)..    a11 * x(i) + a12 * y(i)) =e= b1;
eq2(i)..    a21 * x(i) + a22 * y(i)) =e=  b2;
```

When the index set is $i = \{0, 1, 2\}$ the model is expanded and stacked in the following way:

```
j =e= w1*x(0) + w2*y(0) + w1*x(1) + w2*y(1) + w1*x(2) + w2*y(2)
eq1(0)..    a11 * x(0) + a12 * y(0)) =e= b1;
eq2(0)..    a21 * x(0) + a22 * y(0)) =e= b2;
eq1(1)..    a11 * x(1) + a12 * y(1)) =e= b1;
eq2(1)..    a21 * x(1) + a22 * y(1)) =e= b2;
eq1(2)..    a11 * x(2) + a12 * y(2)) =e= b1;
eq2(2)..    a21 * x(2) + a22 * y(2)) =e= b2;
```

Note that previously we had a model with an objective function and two indexed equations and two variables [x(i) and y(i)] and now we have a model with one objective function, six equations, and six variables [x(0), x(refeq1), x(refeq2), y(0), y(refeq1) and y(refeq2)]. Thus, before solving the model, GAMS transforms a model of n indexed equations into one of n x *card* equations plus the objective function, where *card* indicates the number of elements in the index set. If the index denotes time periods, this is equivalent to transforming a dynamic model with n indexed equations and t time periods into an equivalent static model of $n \times t$ equations plus the objective function.

When, as in Chapters 8 and 13, we are interested in solving a system of equations and not an optimization problem, we just set the objective function equal to any constant value (i.e., j =e= 0;). Thus, when executing the corresponding solver statement,

```
solve model maximizing j using nlp;
```

GAMS expands and stacks the system of equations and it solves it as a by-product of a pseudo-optimization.