COMPUTATIONAL ECONOMICS

Global Warming in GAMS

David A. Kendrick
Ruben P. Mercado
Hans M. Amman

1 Introduction

The basic economics and chemistry of global warming are that an increase in output causes an increase in CO_2 emission, which in turn causes an increase in the concentration of CO_2 in the atmosphere. This increase in CO_2 concentration permits the sun's rays to come into the earth's atmosphere but captures some of them as they are reflected back, thereby increasing the temperature of the earth. The increased temperature results in a decrease in output. Several of the elements in this chain of causation are controversial; however, this simple line of reasoning is a useful place to begin.

In this chapter we use the classic model of Nordhaus (1992) to study the dynamics of global warming. A simple flowchart for that model, reflecting the foregoing comments is shown in Figure 1.

Figure 1: Flowchart



Economic policy can be used to intervene in this cycle. The most common intervention is a *carbon tax*, which raises the price of fossil fuels such as coal, oil, and natural gas and thereby decreases the effective emission of CO_2 and other greenhouse gases. This decreases the CO_2 concentration and thus the temperature, which in turn tends to *increase* output. However, the tax also decreases the efficiency of the economy, thereby producing a tendency to *decrease* output. This trade-off is shown in Figure 2. The basic structure of this dynamic model is one in which the economic externality is a *stock* variable, that is, the CO_2 concentration and the policy variable are used to control a *flow*, namely the CO_2 emissions.

The trade-off was embedded by Nordhaus in a one-sector growth model,

Figure 2: Trade between taxes and efficiency



similar to the **Excel** growth model used earlier in this book, to create an economic model of global warming. However, he developed the model in **GAMS** rather than in **Excel** as is discussed in the following sections. We begin with by presenting the Nordhaus model in mathematics and then turn to a discussion of the same model in **GAMS**.

2 The Mathematical Model

The best place to start is with the production function, which is in the classic Cobb-Douglas form with output produced by capital and labor:

$$Q(t) = \Omega(t)\, A(t)\, K(t)^{\gamma}\, L(t)^{1-\gamma} \tag{1}$$

where

$$\begin{aligned}
Q(t) &= \text{output in period } t \\
\Omega(t) &= \text{climate impacts (see below)} \\
A(t) &= \text{technology in period } t \\
K(t) &= \text{capital in period } t \\
L(t) &= \text{labor force in period } t \\
\gamma &= \text{elasticity of output with respect to capital}
\end{aligned} \tag{2}$$

The unusual aspect of this production function is the presence of the Ω term, which is used to model:

1. the impact of temperature changes on output;
2. the efficiency-loss effects of the carbon tax.

We will see later in more detail that no distinction is made in this model between labor force and population.

We return to a discussion of the Ω term later; however, for now let us move on to the effect of output on greenhouse gas emissions (mostly CO_2), which is modeled with the equation

$$E(t) = [1 - \mu(t)]\, \sigma(t)\, Q(t) \tag{3}$$

where

$$\begin{aligned} E(t) &= \text{greenhouse gas emissions} \\ \mu(t) &= \text{emission control rate – the fractional reduction of emissions} \\ \sigma(t) &= \text{ratio of greenhouse gas emissions to output} \end{aligned}$$

The μ variable is the percentage of greenhouse gas emissions that is prevented from entering the atmosphere. Thus it might be thought of as the action of devices to reduce the CO_2 in the smoke from power plant stacks or to sequester the carbon underground or underwater before it enters the atmosphere. Alternatively, it can be viewed as a proxy for a carbon tax that reduces the use of fossil fuels and thereby the effective emissions.

Next consider the effect of the emissions on the CO_2 concentration in the atmosphere, which is modeled with the equation

$$M(t) = \beta E(t) + (1 - \delta_M)\, M(t-1) \tag{4}$$

where

$$\begin{aligned} M(t) &= CO_2 \text{ concentration relative to preindustrial times} \\ \beta &= \text{marginal atmospheric retention ratio} \\ \delta_M &= \text{rate of transfer from the rapidly mixing reservoirs to the deep ocean} \end{aligned}$$

The two parameters in this equation (β and δ_M) divide the nonintervention optimist from the intervention pessimist on global warming. The β parameter is the proportion of emissions that add to the CO_2 concentration in the atmosphere. The δ_M parameter is a measure of the atmosphere's ability to break down the CO_2. If δ_M is large, then the decay rate of CO_2 in the atmosphere is high and that mitigates the effect of higher emission rates. So the optimists like to believe that β is small and δ_M is large.

The increase in the atmospheric concentration of CO_2 in equation (4) in turn drives changes in the temperature. This is done in two steps in the model. In the first step the increase in atmospheric concentration Mincreases the forcing term F in the equation

$$F(t) = 4.1 \left[\frac{\log \frac{M(t)}{590}}{\log 2} \right] + FO(t) \tag{5}$$

where

$$\begin{aligned} F(t) &= \text{forcing term of greenhouse gas concentration on temperature} \\ FO(t) &= \text{exogenous forcing from other greenhouse gases} \end{aligned}$$

This first term on the right-hand side of equation 5) models the effect of the CO_2 concentration on the forcing term. The equation also includes a separate exogenous term for the effects of all other greenhouse gases on that term.

The forcing term then influences the temperature. However, the temperature is broken into two separate variables in this model:

1. the temperature of the atmosphere and upper oceans;
2. the temperature of the deep oceans.

For simplicity of exposition, we refer to the first of these two as just the temperature of the atmosphere, but the reader should keep in mind that it is actually the temperature of the atmosphere and the upper oceans. The forcing term F drives the temperature of the atmosphere; moreover, the two temperatures have feedback effects on one another. The expression for the temperature of the atmosphere is

$$\begin{aligned} T_1(t) = {} & T_1(t-1) + \\ & \left(\frac{1}{R_1}\right) \left\{ F(t) - \lambda T_1(t-1) - \left(\frac{R_2}{\tau_2}\right) \left[T_1(t-1) - T_2(t-1) \right] \right\} \end{aligned} \tag{6}$$

where

$$
\begin{aligned}
T_1(t) &= \text{temperature of the atmosphere and upper oceans} \\
T_2(t) &= \text{temperature of the deep oceans} \\
R_1 &= \text{thermal capacity of the atmosphere and upper oceans} \\
R_2 &= \text{thermal capacity of the deep oceans} \\
F(t) &= \text{radiative forcing in the atmosphere from greenhouse gases} \\
\lambda &= \text{the climate feedback parameter} \\
\frac{1}{\tau_2} &= \text{the transfer rate from the upper to the lower layer}
\end{aligned}
$$

This function appears complicated at first; however, taking it piece by piece makes it easier to understand. Consider first a simpler version of equation (6) with only the lagged T_1 term and the F term, that is,

$$T_1(t) = T_1(t-1) + \left(\frac{1}{R_1}\right)\{F(t) - \lambda T_1(t-1)\} \tag{7}$$

This is just a dynamic equation of the temperature of the atmosphere driven by the forcing term and mitigated by the climate feedback parameter λ. The other term in equation (7) is the difference between the atmospheric temperature T_1 and the deep-ocean temperature T_2:

$$-\left(\frac{R_2}{\tau_2}\right)[T_1(t-1) - T_2(t-1)] \tag{8}$$

Thus, because of the negative sign in front of the term in equation (8), the greater the difference between the two temperatures the less the atmospheric temperature increases from one period to the next. Thus if the deep ocean is much cooler than the atmosphere it absorbs heat, which results in less increase in the atmospheric temperature. This can also be seen in the equation for the temperature of the deep oceans:

$$T_2(t) = T_2(t-1) + \left(\frac{1}{R_2}\right)\left\{\left(\frac{R_2}{\tau_2}\right)[T_1(t-1) - T_2(t-1)]\right\} \tag{9}$$

In this case there is a positive effect of the temperature difference between the two layers. Thus an increase in the difference between the atmospheric and deep-ocean temperatures in period $t-1$ results in a more rapid increase in the deep-ocean temperature in period t.

Next we need to close the loop of causation in the model from temperature back to output. First, recall the use of the Ω term in the production function in equation (1):

$$Q(t) = \Omega(t)\, A(t)\, K(t)^{\gamma}\, L(t)^{1-\gamma} \qquad \text{[Equation (1)]}$$

The Ω term in the Nordhaus model is driven by the d variable, which is defined as

$$d(t) = a_1 \left[\frac{T_1(t)}{3}\right]^2 \tag{10}$$

$d(t)$ = fractional loss of global output from greenhouse warming

a_1 = a constant

Thus, as the temperature of the atmosphere rises, the fractional loss of global output increases in a nonlinear way. The d term, in turn, appears in the denominator of the Ω term as

$$\Omega(t) = \frac{1 - TC(t)}{1 + d(t)} \tag{11}$$

where

$TC(t)$ = fractional cost to global output from greenhouse gas emission controls

Thus as temperature increases, the d term increases, the Ω term decreases, and output declines. The definition of the Ω term also includes the term TC, which represents the efficiency loss in output that is caused by the imposition of carbon taxes. This loss is represented in the model with the equation

$$TC(t) = b_1 \mu(t)^{b_2} \tag{12}$$

Thus as the carbon tax increases and μ, the fractional reduction of emissions, increases the efficiency loss term TCincreases; also, from equation (11), as this loss increases the Ω term decreases.

In summary, then, the Ω term is indirectly affected by two variables, μ and T, both of which cause it to fall as they increase. The first variable is the fractional reduction of emissions, μ, operating through the TC variable, and the second is temperature, T, operating through the d variable. However, in the model, μ and T are related to one another in an inverse fashion. As the carbon tax underlying μ increases, the temperature T declines. This is the essential trade-off in the model—higher carbon taxes reduce emissions, decrease temperature, and increase output; however, they also impose efficiency loss on the economy and thus reduce output.

There is also a second basic trade-off at work here, which comes from the fact that this is basically a one-sector growth model of the Ramsey type that was modeled in Excel earlier in this book. The trade-off in the growth model is between consumption and investment and is embodied in the equations

$$Q(t) = C(t) + I(t) \tag{13}$$
$$K(t) = (1-\delta_K) K(t-1) + I(t) \tag{14}$$

where

$$\begin{aligned} C(t) &= \text{total consumption in period } t \\ I(t) &= \text{investment in period } t \\ \delta_K &= \text{rate of depreciation of the capital stock} \end{aligned}$$

Thus as consumption rises investment must fall, and as investment falls capital accumulation declines and so output declines. This is in turn linked to the criterion function of the model, which is to maximize discounted utility

$$\max_{[c(t)]} \sum_{t=1}^{T} U[c(t), P(t)](1+\rho)^{-t} \tag{15}$$

where

$$\begin{aligned} U[] &= \text{utility function} \\ P(t) &= \text{population in period } t \\ \rho &= \text{pure rate of social time preference} \\ c(t) &= \text{per capita consumption in period } t \end{aligned}$$

Moreover, the utility in each period is a nonlinear function of per capita consumption (actually, in this model it is consumption per member of the labor force):

$$c(t) = C(t) / L(t) \tag{16}$$

where the utility function is the same general form as was used in the growth model in Excel, that is,

$$U[c(t), L(t)] = L(t)\left\{[c(t)]^{1-\alpha} - 1\right\} / (1-\alpha) \tag{17}$$

In summary, this trade-off is that as total consumption increases it increases per capita consumption and thus utility, but this is achieved by reducing investment and thus capital accumulation, which reduces future output.

This completes the statement of the model. However, since the model is somewhat long it is useful to restate it in summary fashion.

3 The Model in Summary

We begin with the criterion function and continue with the constraints:

Criterion function from (15):

$$\max_{[c(t)]} \sum_{t=1}^{T} U[c(t), L(t)](1+\rho)^{-t} \tag{18}$$

Utility function from equation (17):

$$U[c(t), L(t)] = L(t)\left\{[c(t)]^{1-\alpha} - 1\right\} / (1-\alpha) \tag{19}$$

Production function from equation (1):

$$Q(t) = \Omega(t) A(t) K(t)^{\gamma} L(t)^{1-\gamma} \tag{20}$$

Output division from equation (13):

$$Q(t) = C(t) + I(t) \tag{21}$$

Per capita consumption from equation (16):

$$c(t) = C(t) / L(t) \tag{22}$$

Capital accumulation from equation (14):

$$K(t) = (1 - \delta_K) K(t-1) + I(t) \tag{23}$$

Emissions from equation (3):

$$E(t) = [1 - \mu(t)] \sigma(t) Q(t) \tag{24}$$

CO_2 concentration from equation (4):

$$M(t) = \beta E(t) + (1 - \delta_M) M(t-1) \tag{25}$$

Temperature in the atmosphere and upper oceans from equation (6):

$$T_1(t) = T_1(t-1) + \left(\frac{1}{R_1}\right) \left\{ F(t) - \lambda T_1(t-1) - \left(\frac{R_2}{\tau_2}\right) [T_1(t-1) - T_2(t-1)] \right\} \tag{26}$$

Temperature in the deep oceans from equation (7):

$$T_2(t) = T_2(t-1) + \left(\frac{1}{R_2}\right) \left\{ \left(\frac{R_2}{\tau_2}\right) [T_1(t-1) - T_2(t-1)] \right\} \tag{27}$$

Forcing term from equation (5):

$$F(t) = 4.1 \left[\frac{\log \frac{M(t)}{590}}{\log 2} \right] + FO(t) \tag{28}$$

Fractional loss of output from greenhouse warming from equation (11):

$$d(t) = a_1 \left[{T_1(t)}/{3} \right] \tag{29}$$

Fractional cost to output from controls—carbon taxes from equation (13):

$$TC(t) = b_1 \mu(t)^{b_2} \tag{30}$$

Climate and emission control impact from equation (12):

$$\Omega(t) = \frac{1 - TC(t)}{1 + d(t)} \tag{31}$$

4 The Model in GAMS

The GAMS representation of Nordhaus's DICE model is in the file `dice.gms` and is shown in Appendix C. This implementation of the model uses forty time periods each of which is 10 years long, so the model covers a time horizon of 400 years. It is not uncommon in dynamic models to have more than 1 year per time period, but it does require some adjustments. For example, the capital accumulation equations is changed from

$$K(t) = (1 - \delta_K) K(t-1) + I(t) \qquad \text{[Equation (14)]}$$

to

$$K(t) = (1 - \delta_K)^{10} K(t-1) + 10\, I(t) \tag{32}$$

Since the depreciation rate is annual, it is necessary to raise it to a power that is equal to the number of years per time period. In addition, the flow variables, like investment in this equation, are in annual terms and must be multiplied by the number of years per time period in order to use them appropriately in accumulation equations.

Moreover, some of the other equations in the GAMS statement of the model are in a slightly different form than in the mathematics used earlier. In particular, the production function and the emission equations used in the GAMS statement are obtained by substitution of some equations.

The production function is created by substituting equation (29) and (30) into equation (31) to obtain

$$\Omega(t) = \frac{1 - b_1 \mu(t)^{b_2}}{1 + a_1 \left[{T_1(t)}/{3} \right]^2} \tag{33}$$

or

$$\Omega(t) = \frac{1 - b_1 \mu(t)^{b_2}}{1 + \left({a_1}/{9} \right) T_1(t)^2} \tag{34}$$

Then equation (34) is substituted into equation (1)

$$Q(t) = \Omega(t) A(t) K(t)^{\gamma} L(t)^{1-\gamma} \qquad \text{[Equation (1)]}$$

to obtain

$$Q(t) = \frac{1 - b_1 \mu(t)^{b_2}}{1 + \left(a_1/9\right) T_1(t)^2} A(t) K(t)^{\gamma} L(t)^{1-\gamma} \tag{35}$$

and rearranged to obtain

$$Q(t) = A(t) L(t)^{1-\gamma} K(t)^{\gamma} \left[\frac{1 - b_1 \mu(t)^{b_2}}{1 + \left(a_1/9\right) T_1(t)^2} \right] \tag{36}$$

which is the form of the production function used in the GAMS statement. The emissions equation used in the GAMS representation is obtained by using equation (24), that is,

$$E(t) = [1 - \mu(t)] \sigma(t) Q(t) \tag{37}$$

and substituting the production function from equation (19) into it to get

$$E(t) = [1 - \mu(t)] \sigma(t) \Omega(t) A(t) K(t)^{\gamma} L(t)^{1-\gamma} \tag{38}$$

Then equation (38) is rearranged to obtain

$$E(t) = \sigma(t) [1 - \mu(t)] \Omega(t) A(t) L(t)^{1-\gamma} K(t)^{\gamma} \tag{39}$$

and we set $\Omega = 1$ to get

$$E(t) = \sigma(t) [1 - \mu(t)] A(t) L(t)^{1-\gamma} K(t)^{\gamma} \tag{40}$$

This last step of setting $\Omega = 1$ is surprising so you may want to restore a nonunitary Ω to that equation in the GAMS representation. The parameter σ is treated as time varying in the equation

$$\sigma(t) = \sigma_0 e^{g_\sigma(t)} \tag{41}$$

with

$$g_\sigma(t) = \left(g_{\sigma 0}/\delta_a\right) \left(1 - e^{-\delta_a t}\right) \tag{42}$$

where

$$\begin{aligned} \sigma_0 &= \text{initial } CO_2 \text{ -equivalent emission-GNP ratio} \\ g_\sigma(t) &= \text{cumulative improvement of energy efficiency} \\ g_{\sigma 0} &= \text{growth of } \sigma \text{ per decade} \\ \delta_a &= \text{rate of decline of technological change per decade} \end{aligned}$$

The total factor productivity parameter in the production function is treated in a similar fashion with the equations

$$A\left(t\right) = A_0 e^{g_a(t)} \tag{43}$$

with

$$g_a\left(t\right) = \left({}^{g_{a0}}\!/_{\delta_a}\right)\left(1 - e^{-\delta_a t}\right) \tag{44}$$

where

$$\begin{aligned} A_0 &= \text{initial level of total factor productivity} \\ g_a\left(t\right) &= \text{growth rate of productivity from 0 to } T \\ g_{a0} &= \text{initial growth rate of technology per decade} \end{aligned}$$

The rate of growth of the labor force is treated in the same way with the equations

$$L\left(t\right) = L_0 e^{g_L(t)} \tag{45}$$

with

$$g_L\left(t\right) = \left({}^{g_{L0}}\!/_{\delta_L}\right)\left(1 - e^{-\delta_L t}\right) \tag{46}$$

where

$$\begin{aligned} L_0 &= \text{1965 world population in millions} \\ g_L(t) &= \text{growth rate of labor from 0 to } t \\ g_{L0} &= \text{growth rate of population per decade} \end{aligned}$$

Finally the exogenous forcing term for other greenhouse gases is set using the equations

$$\begin{array}{l} FO\left(t\right) = 0.2604 + 0.125\,t - 0.0034\,t^2 \quad \text{when } t < 15 \\ FO\left(t\right) = 1.42 \quad \text{when } t \geqslant 15 \end{array} \tag{47}$$

Thus this term increases in quadratically from 0.2604 to 1.42 over the first 15 years and then remains constant at 1.42.

If you have already read the previous chapters in this book dealing with models in GAMS, particularly the dynamic models, the GAMS representation of the Nordhaus model provided in Appendix C at the end of this chapter will seem like familiar terrain. If you have not read those chapters, you are encouraged do so and to take a look at Appendixes A and B at the end of the book.

5 Results

In the 1992 *Science* article Nordhaus compares five solutions of the model, which are shown in Table 1. The first solution is a *no-controls* result in which μ is set to zero, that is, there is no removal of emissions relative to the uncontrolled level. The second is the full optimal control solution that provides a slight (0.027%) improvement in total discounted utility over the horizon covered by the model.

Table 1: Solutions of the Model

Case	Policy	Base value	Dollar difference	Percent difference
1	No-controls	731.694	0	0.000
2	Optimal policy	731.893	199	0.027
3	Stabilize emissions	726.531	–5163	–0.706
4	Stabilize climate	701.764	–29930	–4.091
5	Geoengineering	735.787	4093	0.559

The third solution is to fix emissions at around 10 percent above the uncontrolled level after 1995. This requires setting μ equal to 0.1 after 1995 and can be implemented in the GAMS statement of the model by using a MIU.FX statement before the SOLVE statement. As is seen in Table 15.1 this results in a decrease in total discounted utility of about 0.7 percent.

A more drastic policy is to stabilize climate as is shown in the fourth solution. This solution limits the temperature increase to 0.2 $^\circ$C per decade after 1985 with an upper limit of a total increase of 1.5 $^\circ$C from 1990, which results in a decrease of about 4 percent in total discounted utility relative to the uncontrolled solution.

The last solution considers the effects of introducing a hypothetical technology that provides costless mitigation of climate change. Examples cited by Nordhaus include shooting smart mirrors into space or seeding the ocean with iron to accelerate carbon sequestration.

6 Experiments

The obvious experiments with this model are to attempt to replicate some of the solutions shown in Table 1; however, there are a number of other experiments of interest. One such experiment is to decrease the size of parameter a_1 in equation (29):

$$d(t) = a_1 \left[\frac{T_1(t)}{3} \right]^2 \qquad \text{[Equation (10)]}$$

This experiment recognizes that there is considerable controversy about the magnitude of the effect of increases in temperature on economic output. In fact, some Russians seem to have concluded that because of the northerly location of most of their country that slight temperature increases might actually result in increases rather than in decreases in national GDP.

Another experiment would be to increase the parameter δ_M in the CO_2 concentration equation

$$M(t) = \beta E(t) + (1 - \delta_M) M(t-1) \qquad \text{[Equation (5)]}$$

to reflect a feeling that the atmosphere is able to break down more of the CO_2 than the original parameter value reflects.

7 Further Reading

As was mentioned earlier, this chapter is based on the article by Nordhaus (1992) in *Science* about the DICE model. That model has the virtue of being relatively simple and is thus useful for this chapter. For a later model see the RICE model by Nordhaus and Boyer (2000). For a model that is used to analyze the costs of CO_2 emissions limits see Manne and Richels (1992). For an alternative to the IPCC CO_2 emission projections see Eckaus (1994). For a general equilibrium model approach to the analysis of reducing carbon emissions see Blitzer, Eckaus, Lahiri, and Meeraus (1992).

For a model that uses the GAMS software and focuses on the role of the developing countries in global warming—particularly India and China—see Duraiappah (1993). For particular reference to the effects of greenhouse gases in agriculture and forestry see McCarl and Schneider (2001). For a discussion of climate policy change after the Kyoto treaty see McKibbin and Wilcoxen (2002).

For an example of the analysis of water pollution control with a GAMS model see Letson (1992). Those interested in environmental models for various sectors of the economy can finds models of the plastics sector in China, the pulp and paper sector in India, the shrimp industry in Thailand, and the livestock sector in Botswana in Duraiappah (2003).

References

Blitzer, C. R., Eckaus, R. S., Lahiri, S. and Meeraus, A.: 1992, The potential for reducing carbon emissions from increased efficiency: A general equilibrium methodology, *Indian Economic Review* **27**, 199–214.

Duraiappah, A. K.: 2003, *Computational Models in the Economics of Environment and Development*, Advances in Computational Economics, Kluwer Academic Publishers, Dordrecht, the Netherlands.

Eckhaus, R. S.: 1994, Potential CO2 emissions: Alternatives to the ipcc projections, *Mimeo*, MIT, Cambridge, Massachusetts, USA.

Judd, K. L.: 1998, *Numerical methods in economics*, MIT Press, Cambridge, Massachussets.

Letson, D.: 1992, Simulation of a two-pollutant, two-season pollution offsets systems for the colorado river of texas below austin, *Water Resources Research* **28**, 1311–1318.

Manne, A. S. and Richels, R. G.: 1992, *Buying Greenhouse Insurance: The Economic Costs of CO2 Emission Limits*, MIT Press, Cambridge, Massachusetts, USA.

McCarl, B. A. and Schneider, U. A.: 2001, Greenhouse gas mitigation in u.s. agriculture and forestry, *Science* **294**, 2481–2482.

McKibbin, W. J. and Wilcoxen, P. J.: 2002, Climate change policy after kyoto: Blueprint for a realistic approach, *Brookings paper*, Brookings Institution, Cambridge, Massachusetts, USA.

Miranda, M. J. and Fackler, P. L.: 2002, *Applied Computational Economics and Finance*, MIT Press, Cambridge, Massachussets.

Nordhaus, W. D.: 1992, An optimal transition path for controlling greenhouse gases, *Science* **258**, 1315–1319.

Nordhaus, W. D. and Boyer, J.: 2000, *Warming the World: Economic Modeling of Global Warming*, MIT Press, Cambridge, Massachussets, USA.

Appendices

A Nonlinear Optimization Solvers

Solving nonlinear optimization problems usually requires the use of numerical methods. In general, those methods consist of a *smart* trial-and-error algorithm that is a finite sequence of computational steps designed to look for convergence to a solution. There is a variety of algorithms to solve nonlinear problems. Some of them are global methods, in the sense that they perform a parallel exploration of many regions of the optimization space, for example, the genetic algorithm. Other are local, as they tend to focus on the exploration of a particular region of the optimization space. We introduce here two of the most popular local methods—the gradient and the Newton methods—used by the solvers in Excel, **GAMS**, and **MATLAB**, but before introducing them, we give with a simple example.

Suppose that we are trying to find the maximum of a nonlinear function

$$y = f(x) \tag{A-1}$$

such as the one represented in Figure 3.

A simple and very rudimentary algorithm to find the solution might be as follows: We choose an arbitrary initial value for x, such as x_0 in Figure 3, and compute the corresponding $y_0 = f(x_0)$. We then increase that value by a constant magnitude h (we name this magnitude the *search step*) that we also choose in an arbitrary way. For the new value of x, that is x_1, we compute the corresponding value of

$$y_1 = f(x_1) = f(x_0 + h) \tag{A-2}$$

and compare this value to the one obtained in the previous step. We continue to do this as long as the differences between two successive values of y are positive (negative for a minimization problem). As soon as we compute a difference with a negative sign (in Figure F.1 this would correspond to x_2), we reverse the direction of the search. We begin to move in the opposite direction along x (i.e., subtracting h from x) and use a smaller value for h than the one we were using while we moved in the opposite direction. We continue like this until we again find a difference between two successive values of y that is negative, at which point we again reverse the direction of the search and further

Figure 3: A nonlinear function.



reduce the size of h, and so on. We stop when the difference between two successive values of y falls below a preestablished tolerance limit.

The gradient and the Newton methods are iterative like the one just presented. However, they exploit local information about the form of the function; that is, they use the function's derivatives. To illustrate this we change to a multivariate example. In this case we use the following equation to obtain each new value of the vector x:

$$x_{n+1} = x_n + h\Delta x \tag{A-3}$$

where h is the search step—now always a positive value—and Δx is the direction of change, which, as we will see, is determined by the function's derivatives.

The gradient method uses the first derivative or gradient, which gives us information about how the function changes in the neighborhood of a given point. Its basic framework is the well-known first-order Taylor approximation,

$$f(x_{n+1}) \cong f(x_n) + h\nabla f(x_n)\Delta x \tag{A-4}$$

where $\nabla f(x_n)$ is the gradient vector. Note that since h is supposed to be positive, the best direction of motion is

$$\Delta x = \nabla f(x_n) \tag{A-5}$$

for a maximization problem, since

$$f(x_{n+1}) \cong f(x_n) + h\nabla(f(x_n))^2 > f(x_n) \tag{A-6}$$

In addition, for a minimization problem

$$\Delta x = -\nabla f(x_0) \tag{A-7}$$

since

$$f(x_{n+1}) \cong f(x_n) - h\nabla(f(x_n))^2 < f(x_n) \tag{A-8}$$

The basic framework of the Newton method is the second-order Taylor approximation

$$f(x_{n+1}) \cong f(x_n) + h\nabla f(x_n)\Delta x + \frac{h}{2}\Delta x' H(x_n)\Delta x \tag{A-9}$$

where $H(x_0)$ is the second-order derivative or Hessian, which tells us how the slope of the function changes in a neighborhood of a given point.

Assuming the Taylor expansion of a function f is a good global approximation of that function, we approximate the optimum value of f by optimizing its Taylor expansion. In our case, this is equivalent to saying that to determine the best direction of motion Δx we have to optimize the expression (f.9). Differentiating (f.9) with respect to Δx, setting the result equal to zero, and solving for Δx, we obtain

$$\Delta x = -\frac{\nabla f(x_n)}{H(x_n)} \tag{A-10}$$

which is the best direction of motion for the Newton method.

Sometimes iterative methods like the ones presented here do not converge to a solution after a finite number of iterations. This problem can be overcome by changing the maximum number of iterations, or the size of the search step, or the tolerance limit, or the initial value of the search. Most solvers allow you to change these parameters.

Note also, as is the general case for numerical methods dealing with nonlinear optimization problems, that if there is more than one local optimum we will find only one of them. Thus, we never know for sure if the optimum we reached was a local or a global one. A rough way of dealing with this difficulty is to solve the problem providing the algorithm with alternative initial values of the search.

In this appendix we presented three numerical methods of increasing complexity. Of course, the more complex ones make use of more information, thus reducing, in general, the number of steps needed to achieve convergence. However, those steps become more complex, since they require the computation of a gradient or a Hessian. Then there are trade-offs to be evaluated when choosing a solution method.

There are additional methods for solving nonlinear problems numerically—for example, the conjugate gradient method, the penalty function method, and the sequential quadratic programming—a number of which extend, combine, or mimic the ones introduced here. For a comprehensive presentation refer to Judd (1998) and Miranda and Fackler (2002). The Excel solver uses a conjugate gradient method or a Newton method. **GAMS** uses a variety of methods, depending on what you choose or have set up as the default nonlinear solver. The **MATLAB** solver used in Chapter 7 and invoked by the fmincon function uses a sequential quadratic programming method. For details on the specific methods used by **Excel**, **GAMS**, and **MATLAB** refer to their corresponding user's and solver's manuals.

B The Stacking Method in GAMS

As a compact way of expressing a multiequation model, GAMS allows us to write indexed equations. As seen in several of the chapters in this book, those indices may represent commodities, locations, time periods, and so on.

For example, the equations corresponding to a problem such as

$$\max J = \sum_{i=0}^{2} w_1 x_i + w_2 y_i \tag{B-1}$$

subject to the constraints

$$a_{11}x_i + a_{12}y_i = b_1 \tag{B-2}$$
$$a_{21}x_i + a_{22}y_i = b_2 \tag{B-3}$$

can be represented in GAMS as

```
eqj..       j =e= sum(i, w1 * x(i) + w2 * y(i));
eq1(i)..    a11 * x(i) + a12 * y(i)) =e= b1;
eq2(i)..    a21 * x(i) + a22 * y(i)) =e=  b2;
```

When the index set is $i = \{0, 1, 2\}$ the model is expanded and stacked in the following way:

```
j =e= w1*x(0) + w2*y(0) + w1*x(1) + w2*y(1) + w1*x(2) + w2*y(2)
eq1(0)..    a11 * x(0) + a12 * y(0)) =e= b1;
eq2(0)..    a21 * x(0) + a22 * y(0)) =e= b2;
eq1(1)..    a11 * x(1) + a12 * y(1)) =e= b1;
eq2(1)..    a21 * x(1) + a22 * y(1)) =e= b2;
eq1(2)..    a11 * x(2) + a12 * y(2)) =e= b1;
eq2(2)..    a21 * x(2) + a22 * y(2)) =e= b2;
```

Note that previously we had a model with an objective function and two indexed equations and two variables [x(i) and y(i)] and now we have a model with one objective function, six equations, and six variables [x(0), x(refeq1), x(refeq2), y(0), y(refeq1) and y(refeq2)]. Thus, before solving the model, GAMS transforms a model of n indexed equations into one of *n x card* equations plus the objective function, where *card* indicates the number of elements in the index set. If the index denotes time periods, this is equivalent to transforming a dynamic model with n indexed equations and t time periods into an equivalent static model of $n \times t$ equations plus the objective function.

When, as in Chapters 8 and 13, we are interested in solving a system of equations and not an optimization problem, we just set the objective function equal to any constant value (i.e., j =e= 0;). Thus, when executing the corresponding solver statement,

```
solve model maximizing j using nlp;
```

GAMS expands and stacks the system of equations and it solves it as a by-product of a pseudo-optimization.

C The GAMS Representation of the Global Warming Model

```
$offsymxref offsymlist
* Explaining the DICE, Cowles Foundation Discussion Paper, January 1991
* The calibration is to a 60-period run for the transversality
*
sets    t               time periods     /1*40/
        tfirst(t)       first period
        tlast(t)        last period

scalars bet     elasticity of marginal utility          /0/
        r       rate of social time preference per year /0.03/
        gl0     growth rate of population per decade    /0.223/
        dlab    decline rate of population growth per decade    /0.195/
        deltam  removal rate carbon per decade  /0.0833/
        ga0     initial growth rate for technology per decade   /0.15/
        dela    decline rate of technological change per year   /0.11/
        sig0    co2-equivalent emissions-gnp ratio      /0.519/
        gsigma  growth of sigma per decade      / - 0.1168 /
        dk      depreciation rate on capital per year /0.10/
        gama    capital elasticity in production function       /0.25/
        m0      co2-equivalent concentrations 1965 billions t c /677/
        tl0     lower stratum temperature (c) 1965      /0.10/
        t0      atmospheric temperature (c) 1965        /0.2/
        atret   marginal atmosphere retension rate      /0.64/
        q0      1965 world gross output trillion 89 US$    /8.519/
        ll0     1965 world population million   /3369/
        k0      1965 value capital trillion 1989 US$    /16.03/
        c1      climate-equation coefficient for upper level    /0.226/
        lam     climate feedback factor /1.41/
        c3      transfer coefficient upper to lower stratum     /0.440/
        c4      transfer coefficient for lower level    /0.02/
        a0      initial level of total factor productivity      /0.00963/
        a1      damage coefficient for co2 doubling (fraction GWP)      /0.0133/
        b1      intercept control cost function /0.0686/
        b2      exponent of control cost function       /2.887/
        phik    transversality coeff capital ($ per unit)        /140/
        phim    transversality coeff carbon ($ per unit)  / - 9.0 /
        phite   transversality coeff temperature ($ per unit)  / - 7000 /

parameters      l(t)    level of population and labour
                al(t)   level of total factor productivity
                sigma(t)        co2-equvalent-emissions output ratio
                rr(t)   discount factor
                ga(t)   growth rate of productivity from 0 to t
                forcoth(t)      exogenous forcing for other greenhouse gases
                gl(t)   growth rate of labour 0 to t
                gsig(t) cumulative improvement of energy-efficincy
                dum(t)  dummay variable 0 except last period ;

        tfirst(t) = yes$(ord(t) eq 1);
        tlast(t)  = yes$(ord(t) eq card(t));
        display tfirst, tlast;

        gl(t) = (gl0/dlab)*(1-exp(-dlab*(ord(t)-1)));
        l(t)  = ll0*exp(gl(t));
        ga(t) = (ga0/dela)*(1-exp(-dela*(ord(t)-1)));

        al(t) = a0*exp(ga(t));
```

```
        gsig(t) = (gsigma/dela)*(1-exp(-dela*(ord(t)-1)));
        sigma(t) = sig0*exp(gsig(t));

        dum(t) = 1$(ord(t) eq card(t));
        rr(t)  = (1+r)**(10*(1-ord(t)));

        forcoth(t)  =  1.42;
        forcoth(t)$(ord(t) lt 15) = 0.2604 + 0.125*ord(t) - 0.0034*ord(t)**2;

variables       miu(t)  emission control rate GHGs
                forc(t) radiative forcing, W per m2
                te(t)   temperature, atmosphere C
                tl(t)   temperature, lower ocean C
                m(t)    co2 equivalent concentration bill t
                e(t)    co2 equivalent emissions bill t
                c(t)    consumption trillion US$
                k(t)    capital stock trillion US$
                cpc(t)  per-capita consumption 1000s US$
                pcy(t)  per-capita income 1000s US$
                i(t)    investment trillion US$
                s(t)    savings rate as fraction of GWP
                ri(t)   real interest rate per annum
                trans(t)        transversality variable last period
                y(t)    output

                utility;

positive variables miu, e, te, m, y, c, k, i;

equations       util    objective function
                yy(t)   output equation
                cc(t)   consumption equation
                kk(t)   capital balance equation
                kk0(t)  initial condition for k
                kc(t)   terminal condition for k
                cpce(t) per-capita consumption definition
                pcye(t) per-capita income definition
                ee(t)   emissions precess
                seq(t)  savings rate equation
                rieq(t) interest rate equation
                force(t)  radiative forcing equation
                mm(t)   co2 distribution equation
                mm0(t)  initial condition for m
                tte(t)  temperature-climate equation for atmosphere
                tte0(t)  initial condition for atmospheric temperature
                tle(t)  temperature-climate equation for lower oceans
                transe(t) transversality condition
                tle0(t) initial condition for lower ocean ;

* Equations of the model

kk(t).. k(t+1) =l= (1-dk)**10*k(t) + 10*i(t) ;
kk0(tfirst)..   k(tfirst) =e= k0 ;
kc(tlast)..     r*k(tlast) =l= i(tlast) ;

ee(t).. e(t) =g= 10*sigma(t)*(1 - miu(t))*al(t)*l(t)**(1 - gama)*k(t)**gama ;
force(t)..      forc(t) =e= 4.1*(log(m(t)/590)/log(2)) + forcoth(t) ;
mm0(tfirst)..   m(tfirst) =e= m0 ;
mm(t+1).. m(t+1) =e= 590 + atret*e(t) + (1 - deltam)*(m(t) - 590) ;

tte0(tfirst)..  te(tfirst) =e= t0 ;
tte(t+1)..      te(t+1) =e= te(t)+c1*(forc(t)-lam*te(t)-c3*(te(t)-tl(t))) ;
```

```
tle0(tfirst)..  tl(tfirst) =e= tl0 ;
tle(t+1)..      tl(t+1) =e= tl(t) + c4*(te(t) - tl(t));

yy(t).. y(t) =e= al(t)*l(t)**(1-gama)*k(t)**gama*(1-b1*(miu(t)**b2))/(1+(a1/9)*sqr(te(t)));
seq(t)..        s(t) =e= i(t)/(.001+y(t)) ;
rieq(t)..       ri(t) =e= gama*y(t)/k(t) - (1-(1-dk)**10)/10 ;

cc(t).. c(t) =e= y(t) - i(t) ;
cpce(t)..       cpc(t) =e= c(t)*1000/l(t) ;
pcye(t)..       pcy(t) =e= y(t)*1000/l(t) ;

transe(tlast).. trans(tlast) =e= rr(tlast)*(phik*k(tlast)+phim*m(tlast)+phite*te(tlast));

util..  utility =e= sum(t,10*rr(t)*l(t)*log(c(t)/l(t))/0.55+trans(t)*dum(t));

* Upper and lower bounds; general conditions imposed for stability

miu.up(t) = 0.99;
miu.lo(t) = 0.01;
k.lo(t) = 1;
te.up(t) = 20;
m.lo(t) = 600 ;
c.lo(t) = 2;

* Upper and lower bounds for historical constraints

miu.fx('1') = 0.0;
miu.fx('2') = 0.0;
miu.fx('3') = 0.0;

* Solution options

option iterlim = 99999;
option reslim = 99999;
option solprint = off;
option limrow = 0;
option limcol = 0;

model co2 /all/ ;
solve co2 maximising utility using nlp ;

* Display of results

display y.l, c.l, s.l, k.l, miu.l, e.l, m.l, te.l, forc.l, ri.l ;
display cc.m, ee.m, kk.m, mm.m, tte.m, cpc.l, tl.l, pcy.l, i.l ;
display sigma, rr, l, al, dum, forcoth ;
```